

Array: Construction and Usage of Arrays of Macro Variables

Ronald Fehd

Centers for Disease Control, and Prevention, Atlanta GA USA

ABSTRACT

The SAS[®] software data step statement `array V (3) $ V1-V3 ('A' 'B' 'C');` produces three character variables named V1, V2, and V3 with corresponding initial values, 'A', 'B', and 'C' and a function, `dim(V)` which returns a value of 3. Programmers can write simple macro tools for use in larger macro procedures. These tools can duplicate SAS software data step constructions that the programmer is comfortable using and make reading and comprehension easier. The macro statement `%Array(V,A B C)` produces three macro variables, V1, V2, and V3, with corresponding values: A, B, and C and macro variable `Dim_V` with the value 3. These variables can then be used in the macro iterative loop statement `%Do I = 1 %to &Dim_V.;` This paper examines the SAS data step array statement and discusses the issues in constructing and using arrays of macro variables. The macro `Array` takes parameters of either a list of elements or a data set.

Macro `Array` is a basic utility used in two other macros that address analysis of multiple-response data. See [3], [4].

INTRODUCTION

A common task for an experienced programmer is to recognize a recurring pattern of code and encapsulate that pattern in a routine which simplifies the processing presentation, while still enabling later readers of the program to grasp the complex concepts that have been coded.

The SAS software macro language is a simple yet powerful programming language. This article examines the SAS software array and associated do loop statements with the idea of translating those concepts into SAS software macro language usage.

SAS ARRAY STATEMENT AND THE DIMENSION (DIM) FUNCTION

1. `array`: SAS statement key-word
2. `array name`: required
3. `subscript`: required, either of a number or asterisk, in parenthesis
 - (3): a number supplied for creation of a series of variables
 - (*): asterisk indicating subscript is determined by SAS software by counting the supplied array-elements
4. `array elements`: an optional list of variable names

The explicit array statement in SAS software has seven phrases; we will examine the four that are most commonly used:

The dimension function — `dim(ABC)` — has two parts:

1. `dim` is the SAS function name
2. `parameter` is an array name defined in the same data step

A typical usage of the array statement would consist of accessing one set of variables in order to repeat some processing on each variable. The example in Program 1 below reads in three Fahrenheit temperatures and converts them to Celsius. Note that the PROC Contents listing shows that SAS has created a series of variables based on the absence of the array-elements in the `array Celsius` statement. Their names — `Celsius1`, `Celsius2`, and `Celsius3` — correspond to the way the variables are accessed in the iterative loop by the array convention of `Celsius(I)`, i.e., `Celsius(1)`, `Celsius(2)`, and `Celsius(3)`.

```

1  DATA  Temperatures;
2  array Celsius  {3};*no array elements,
3          see Contents;
4  array Farnheit  {*} Low Med Hi;
5  drop  I;
6  input Low Med Hi;
7  do I = 1 to dim(Farnheit);
8      Celsius{I} = (Farnheit{I}-32) * 5/9;
9  end;
10 cards;*<deleted for brevity>;
11 PROC Contents;run;

```

Program 1 output				
#	Variable	Type	Len	Pos
-	-----	----	---	---
4	Celsius1	Num	8	24
5	Celsius2	Num	8	32
6	Celsius3	Num	8	40
3	Hi	Num	8	16
1	Low	Num	8	0
2	Med	Num	8	8

SAS SOFTWARE MACRO LANGUAGE ITERATIVE LOOP

To replicate the SAS software iterative loop in the macro language we use a sequentially numbered series of macro variables and a macro variable containing the dimension:

The macro iterative loop and usage of the macro variables can then be written in a form that is visually similar to the SAS software iterative loop.

```

1  %Let Var1 = Q04A;
2  %Let Var2 = Q04B;
3  %Let Var3 = Q04C;
4  %Let Dim_Var = 3;
5
6  %Do I = 1 %to &Dim_Var.;
7      %Put Var&I.  :: &&Var&I.; %end;

```

This loop writes the following note to the SAS log:

```

Program 2 output
Var1  :: Q04A
Var2  :: Q04B
Var3  :: Q04C

```

This is a construction used regularly in certain types of macros. The purpose of this paper is to construct a macro that supports this iterative loop. Such a macro would be named `Array`, and would have two of the SAS array statement phases as parameters: array name, and array-element values. This macro would return a sequentially-numbered series of macro variables and the dimension of the array. The array-element values could be either a provided list or the values of a variable in a data set. This second option of providing the array-element values in a data set would enable macro procedures to be completely data-driven. A `where` parameter has been added to facilitate subsetting of data sets used as parameters. See [3], [4] for examples.

PARAMETERS AND CONSTRAINTS

The simplicity of the macro language both allows and requires construction of a routine that has the appearance of the SAS software `array` statement. Since this is a routine and not a SAS software implementation, there are relations among the parameters that are constraints.

The first and most obvious is that the array-name parameter must follow SAS naming conventions. SAS names may be up to 32 characters in length. For this routine, some number of characters must be reserved for the sequential numbering of the suffix. As the magnitude of the number of array elements increases, the length of the array name must decrease in order for the combined length to be less than or equal to 32.

A second constraint on the array name parameter is that the macro variable used for the dimension has the form: `Dim_<array-name>`. This construction was chosen to appear visually similar to the usage of the dimension function: `dim(<array-name>)`. This convention reduces the length of the array-name as suffix to 28 characters.

The array-name parameter is both prefix and suffix. As suffix to the name of the returned value of dimension, it can be no more than 28 characters in length. As prefix to the series of macro variables 28 characters in the array name allows a maximum of 9,999 sequentially numbered macro variables to be created without suffering a `SAS name too long` error. For larger arrays, the length of the array name can be as small as one character.

Array elements in the SAS software data step array statement are assumed to be delimited by spaces. When array-element values are provided to this routine as a list, the macro `%scan` function is used to pick out each value. The delimiters of the macro function `%scan` are the set of non-alpha-numeric characters. For special cases where, for instance, an array-element value may contain two or more words, the delimiter parameter may be supplied.

A data set and variable name may be supplied as parameters, instead of a list. This routine was written to handle various series of variable names, which were subsets of a `PROC Contents` output data set. Review the test data supplied with the macro.

CASE 1: SCANNING MACRO VALUES FROM A LIST

The macro function %scan operates the same as the SAS software function. In order to construct a loop which has a data-dependent termination, it is necessary to use and test a temporary variable for the exit condition. Here is pseudo-code and macro statements for a loop that converts a list to array elements:

```
initialize: I := 1                                %Let I = 1;
           get I-th Item from List                %Let Item      = %scan(&List,&I);
loop:      %Do %until(&Item. = );
           assign Item                            %Let &Name.&I = &Item.;
           increment I                            %Let I      = %eval(&I + 1);
           get I-th Item from List                %Let Item      = %scan(&List,&I);
until     Item is blank                          %end;
assign dimension                               %Let Dim_&Name. = %eval(&I - 1);
```

Whereas the pseudo-code shows that the test is done at the bottom of the loop, SAS attaches the %until function to the iterative %Do at the top of the loop. The index is incremented using the %eval function. At the loop exit the index is off by one; the dimension is therefore `index - 1`.

CASE 2: SQL SELECT INTO MACRO VALUES FROM A DATA SET VARIABLE

The SQL select into statement can allocate as many macro variables as there are rows in the data set. Here is an example of the basic syntax:

This usage replaces the symput function which was used to transfer values from a data set variable to the macro environment. See [2] for an explanation of symput usage to create a macro array.

```
PROC SQL noprint;
  select <variable name>
  into   :Name1 - :Name9999
  from   <<libref.>><data set name>
  ;quit;
%Let   Dim_Name = &SqlObs.;
```

Note that the test data includes an SQL case statement and an example of passing real numbers into a macro array.

USAGE OF %ARRAY IN OTHER MACROS

The code for creating a macro array from a list was first written as part of the %CheckAll macro. This macro analyzes multiple-response data, a series of variables which contain answers to survey questions with the instructions 'check all that apply'. After typing in hundreds of variables as lists for the various series, I wrote the second section which uses a previously prepared subset of a PROC Contents data set. This addition allows both research and production usage of the %CheckAll macro. See Fehd [3], [4] and test data with the macro.

Dilorio [1] discusses macro arrays of data set names.

CONCLUSION

The SAS software array and do statements are a simple programming tool which allow a programmer to access a list of variables. The macro language allows a programmer to access a list of items with a %Do; statement but lacks a specific %Array statement. This paper has presented a macro array which converts either a list or values of a variable into a sequentially-numbered series of macro variables with common prefix and sequential numeric suffix and also returns a macro variable with the dimension. This routine hides complexity and simplifies readability of programs which contain macro loops.

The SAS software macro language is a simple language. Its simplicity leaves many advanced programming concepts apparently unavailable. Its simplicity is an asset in that, with some forethought and planning, generic tools can be relatively easily written. This macro was initially developed to take a list of variable names as a parameter. After some usage it became apparent that adding the option to accept a data set as parameter would eliminate tedious typing of the variable lists, and, in addition, since the routine was then data-driven, guarantee the accuracy of the data thus processed.

REFERENCES

- [1] Dilorio, Frank (1996), *MACArray: a Tool to Store Dataset Names in a Macro 'Array'*, Proceedings of the Fourth Annual Conference of the SouthEast SAS Users Group, 1997.
- [2] Fehd, Ronald (1997), *%Array: construction and usage of arrays of macro variables*, Proceedings of the 22nd Annual SAS[®] Users Group International Conference, 1997.
<http://www2.sas.com/proceedings/sugi22/CODERS/PAPER80.PDF>
SUGI 29: 2004 Montreal Quebec Canada Coder's Corner
<http://www2.sas.com/proceedings/sugi29/070-29.pdf>
- [3] Fehd, Ronald (1997), *%CHECKALL, a macro to produce a frequency of response data set from multiple-response data* Proceedings of the 22nd Annual SAS[®] Users Group International Conference, 1997.
<http://www2.sas.com/proceedings/sugi22/POSTERS/PAPER236.PDF>
- [4] Fehd, Ronald (1997), *%ShowComb: a macro to produce a data set with frequency of combinations of responses from multiple-response data* Proceedings of the 22nd Annual SAS[®] Users Group International Conference, 1997. <http://www2.sas.com/proceedings/sugi22/POSTERS/PAPER204.PDF>

SAS[®] is a registered trademark of SAS Institute, Inc. In the USA and other countries, ® indicates USA registration.

To receive the latest edition of the macro Array send an e-mail to the author with the subject: request Array.

Author: Ronald Fehd **e-mail: Ronald.Fehd@cdc.hhs.gov**
Centers for Disease Control MS-G23 **e-mail: RJF2@cdc.gov**
4770 Buford Hwy NE
Atlanta GA 30341-3724 **bus: 770/488-8102**

End note: This paper was typeset in L^AT_EX. For further information about using L^AT_EX to write your SUG paper, consult the SAS-L archives:

<http://www.listserv.uga.edu/cgi-bin/wa?S1=sas-l>
Search for :
The subject is or contains: LaTeX
The author's address : RJF2
Since : 01 June 2003

```

_____ array.sas: description _____
1
2  /*      name: Array.sas
3      SAS type: macro, procedure
4  program type: sub-routine
5      purpose: provide data structure for macro loop: %do I = 1 %to &dim
6  description: given a list of values
7              returns a set of macro-variables
8              named &Name.1 &Name.2 .. &Name.N
9              and a macro-variable named Dim_&Name.
10             i.e. %Array(Var,Q04A Q04B Q04C);
11             returns: Var1::Q04A Var2::Q04B Var3::Q04C, Dim_Var::3
12  input   : required: array name      , is macro variable prefix
13             required: array elements, macro variable values, either of
14             horizontal list of values
15             vertical  list: data set and variable
16             option: make macro variables global
17             option: recode variables using Var = 'text' !! Var
18  process : case 1: %scan list of elements
19             case 2: data set: SQL select into macro variables
20             case 3: error msg
21  output  : macro variables: sequentially-numbered series
22             dimension of macro variable array
23  usage   : within macro: see also test data
24             %Array(Array-Name,Item-List);                               run;
25             %Array(Array-Name,Data=Data-Name,Var=Var-Name);           run;
26             : in open code:
27             %Array(Array-Name,Data=Data-Name,Var=Var-Name,_Global_=1);run;
28             : loop: %local I;
29             %Do I = 1 to &Dim_ABC;%put ABC&I &&ABC&I; %end;
30  notes   : scope: must declare %local macro variables Dim_Array-Name
31             and Array-Name1 ... Array-NameN -before- calling Array
32             length(Array-Name) must be <= 28
33             length(Array-Name) + length(Dim_Array-Name) must be <= 32
34  keywords: array %Array Contents delimiter loop macro ParmBuff %scan
35             %Qscan symput SysPbuff %until SQL                          */
36  %MACRO Array /* ----- */
37      (Name      /*m-var array-name: len(array-name) must be <= 28 */
38      ,List      /*horizontal list of array-elements                */
39      ,Data      = . /*vertical list: data set name                  */
40      ,Var       = . /*vertical list: variable in data set          */
41                /* or SQL select phrase                            */
42      ,Dim       = 9 /*vertical list: how many _global_ vars?       */
43      ,Where     = 1 /*vertical list: subset Dictionary.tables      */
44      ,Delimitr  = %str( ).<%str( )]&!$*%str( )^-/%>\/*horiz-list dlm */
45      ,_Global_ = 0 /*?make mac-vars %global?,                      */
46                /* else %local when called by other macro        */
47      ,ShowMsg   = 1 /*write success msg to log?                    */
48      ,Testing   = 0 /*to see SQL select use ,testing=1)           */
49  )/des = "create array of mac-vars from list/data &SysDate."/*
50  /* ** store source /* must have options mStored SASmStore=libref */
51  parmbuff /* see SysPbuff */
52  ;/* ..... */

```

```

53
54 %local I ; %let I = 1; %*macro loop counter;
55 %local Item ; %let Item = .; %*temp var for Qscan;
56 %local MaxLenMvarName; %let MaxLenMvarName = 28; %*max length Name;
57 %local SQLPrint ; %let SQLPrint = noPrint; %*print SQL results;
58 %if &Testing %then %do; %let SQLPrint = print; %end;
59 %if %length(&Name.) gt &MaxLenMvarName. %then %do;%GoTo ErrMsg; %end;
60
61 /*case 1: Name & List */
62 %If "&Name." ne ""
63 and "&List." ne "" %then %do;
64 %If &_Global_ %then %do; %GLOBAL Dim_&Name.; %end;
65 %let I = 1;
66 %let Item = %Qscan(&List.,&I.,&Delimitr);
67 %Do %until(&Item. = );
68 %If &_Global_ %then %do; %GLOBAL &Name.&I; %end;
69 %let &Name.&I = &Item.;
70 %let I = %eval(&I + 1);
71 %let Item = %Qscan(&List,&I.,&Delimitr);
72 %*Do %until; %end;
73 %let Dim_&Name. = %eval(&I. - 1);
74 %If &ShowMsg. %then
75 %Put @@&SysMacroName &Name. dim=&&Dim_&Name. from List<&List.>;
76 %end;%*case 1;
77
78 /*case 2: Name & (Data & Var)*/
79 %Else %If "&Name." ne ""
80 and "&List." eq ""
81 and "&Data." ne "."
82 and %qsubstr(&Var.,1,1) ne .
83 %then %do;
84 %If &_Global_ %then %do; %GLOBAL Dim_&Name.;
85 %LET Dim_&Name. = &Dim.;
86 %Do I = 1 %to &&Dim_&Name.; %GLOBAL &Name.&I.; %end;
87 %*If _Global_; %end;
88 PROC SQL &SQLprint.;
89 select &Var.
90 into :&Name.1 - :&Name.9999
91 from &Data.
92 where &Where.
93 ;quit;
94 %let Dim_&Name. = &SqlObs.;
95 %If &ShowMsg. %then
96 %Put @@&SysMacroName &Name. dim=&&Dim_&Name. from &Data..<&Var.>;
97 %*case 2; %end;
98 /*case 3: print err msg */
99 %Else %do; %ErrMsg:
100 %Put @@ERR%str()OR: in macro &SysMacroName;
101 %Put array-name : required;
102 %Put array-elements may be list, or (data and var);
103 %Put list : Array(Array-Name,Item-List);
104 %Put data+var: Array(Array-Name,Data=Data-Name,Var=Var-Name);
105 %If %length(&Name.) gt &MaxLenMvarName. %then
106 %Put length(Array-Name) must be <= &MaxLenMvarName.;
107 %Put parmlist: <&SysPbuff.>;%* /ParmBuff macro statement option*;
108 %Put parm: Array-Name = <&Name.> length = <%length(&Name.)>;
109 %Put parm: List = <&List.>;
110 %Put parm: Data = <&Data.> ;
111 %Put parm: Var = <&Var.> ;
112 %Put parm: _Global_ = <&_Global_> in (0,1);
113 %end; %*Array; %Mend;

```

```

114
115 /*\begin{testSuite:macro} ***** **
116 DATA VQ04;          *convention for %CheckAll setup;
117 length Name $ 8;    *can use PROC Contents output data set;
118 Name = 'Q04A'; output;Name = 'Q04B'; output;Name = 'Q04C'; output; run;
119
120 %MACRO Demo(Test = 0, Dim_Var = 10, Nobs = 0);%local I;
121 %*initialize: declare macro array names;
122 %Do I = 1 %to &Dim_Var;          %local Var&I.;                                %end;
123 %If      &Test. eq 1 %then %do;%Array(Var,Q03A Q03B Q03C Q03D); %end;
124 %Else %If &Test. eq 2 %then %do;%Nobs(Dim_Var, Data=Vq04); run;
125                                     %Array(Var,Data=Vq04,Var=Name ); %end;
126 %Else %If &Test. eq 3 %then %do;%Array(Var,Data = Dictionary.Tables
127                                     ,Var = Memname
128                                     ,Where = Libname eq 'WORK'
129                                     ); %end;
130 %Else %Goto Exit;run;%*illustrate usage in loop;
131 %Do I = 1 %to &Dim_Var.; %Put Var&I. :: &&Var&I.; %end;
132 %Exit: run;%* ..... Demo *; %Mend;
133
134 %Demo(Test = 1);
135 %Demo(Test = 2);
136 %Demo(Test = 3);
137
138 %Macro PutArray(Name=,Dim = &&Dim_&Name.);%local I;
139 %Do I = 1 %to &Dim.;%Put &Name.&I.<&&Name.&I.>;%end;%Mend;
140
141 %Array(Test,A B C D E, _Global_=1);*make mac-vars %global in open code;
142 %PutArray(Name=Test);
143
144 DATA X;X='.';output;X='X';output;stop;
145 %Array(Z,Data = X
146         ,Var = case when X eq 'X' then '1' else '0' end
147         ,_global_ = 1
148         );
149 %PutArray(Name=Z);
150
151 DATA Test;do Denom = 1 to 4;Real = 1/Denom;output;end;stop;
152 %ARRAY(Real,data = Test
153         ,var = put(put(Real, rb8.),$hex16.)
154         ,_global_=1);
155 %PutArray(Name=Real);
156
157 %ARRAY(Func,data = Test
158         ,var = 'input("&&Real2"x , rb8.) !! put(put(Real, rb8.),$hex16.)
159                 !! "&&x , rb8.)'
160         ,_global_=1);
161 %PutArray(Name=Func);
162
163 DATA _Null_;
164 Half = input("&&Real2"x , rb8.);put Half =;
165 Third = input("&&Real3"x , rb8.);put Third =;
166 Fourth= &&Func4 ;put Fourth=;stop;
167 run;
168 %Array(This_Mvar_Name_Is_Tooooo_Long,A B C D E);*See error msg;
169 run;/*\end{testSuite:macro} ..... Test Data End*/

```